
Change 1 – f_EUTRA_InitialRegistration_Step14_15
f_EUTRA_InitialRegistration_Step14_15 is split into a common function f_EUTRA_InitialRegistration_Step14_15_Common, which is independent form the 36.508 specific PDN mapping and the LTE specific EPS bearer contexts, and the LTE specific f_EUTRA_InitialRegistration_Step14_15.
	  function f_EUTRA_InitialRegistration_Step14_15_Common(EUTRA_CellId_Type p_CellId,
                                                        EUTRA_ATTACH_Type p_ForcedAttach,
                                                        template (omit) AdditionalUpdateType p_AdditionalUpdateType,
                                                        ProcedureTransactionIdentifier p_EPS_TI,
                                                        boolean p_IPv4AllocationViaNasFlag,
                                                        template (omit) ProtocolConfigOptions p_PcoToUE,
                                                        AccessPointName p_APN,
                                                        PDN_Index_Type p_PdnIndex,
                                                        EPS_BearerIdentity p_EpsDefaultBearerId,
                                                        BEARER_CONTEXT_TYPE p_BearerContextNumber,
                                                        template (value) GPRS_Timer p_T3412 := cs_GprsTimer_v_deact,
                                                        template (omit) GPRS_Timer p_T3402 := omit,
                                                        template (omit) GPRS_Timer p_T3423 := omit,
                                                        template (omit) PLMN_List p_EquivalentPlmnList := omit,
                                                        integer p_NoOfEmergencyNumbers := 0, // @sic R5-115673 sic@
                                                        template (omit) EMM_Cause p_EMM_Cause := omit,
                                                        template (omit) DCN_ID p_DCN_Id := omit,
                                                        template (omit) EUTRA_NR_PduSessionInfo_Type p_PduSessionInfo := omit,
                                                        template (omit) ExtdProtocolConfigOptions p_ExtdPco := omit) runs on EUTRA_PTC
  {
    var PDN_TypeToBeUsed_Type v_PDN_TypeToBeUsed := f_EUTRA_MobileInfo_GetPdnTypeToBeUsed();   /* @sic R5s170307 sic@ */
    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(p_ForcedAttach); // @sic R5-103673, R5s100795 sic@
    
    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(p_CellId);
    var template (value) MobileIdentity v_Guti := f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams);
    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);
    
    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode(p_CellId);
    var template (omit) LocAreaId v_LAI := f_GetLAI(v_PLMN, v_LAC, p_ForcedAttach); // @sic R5s110176 sic@
    var TrackingAreaCode v_TAC := f_EUTRA_CellInfo_GetTAC(p_CellId);
    
    var template (omit) EmergNumList v_EmergNumList := omit;
    var template (omit) MobileIdentity v_MSId := f_GetMSId(p_ForcedAttach); // @sic R5-103673 sic@
    var template (omit) ESM_Cause v_ESM_Cause := omit;        // @sic R5s120136 sic@
    var template (value) PDN_Address v_PDN_Address;
    var EmergencyNumList v_EmergencyNumList;
    
    if (pc_IPv6 and pc_IPv4) {              /* @sic R5-120234, R5-120541, R5-120542 sic@ */
      select (v_PDN_TypeToBeUsed) {
        case (forceIPv4only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110010'B); }     // PDN type IPv4 only allowed
        case (forceIPv6only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110011'B); }     // PDN type IPv6 only allowed
        case (pdnTypeAsSupportedByUE) { v_ESM_Cause := omit; }
      }
    }
    
    if (p_NoOfEmergencyNumbers > 0) {
      v_EmergencyNumList := f_Get_EmergencyNumList (p_NoOfEmergencyNumbers, f_EUTRA_MobileInfo_GetLocalEmergencyNumberList()); /* @sic R5s141315 sic@ */
      f_EUTRA_MobileInfo_SetLocalEmergencyNumberList (v_EmergencyNumList); // save the list of locally usable emergency numbers
      v_EmergNumList := f_Build_EmergNumList(p_NoOfEmergencyNumbers, v_EmergencyNumList);
    }

    v_PDN_Address := f_EUTRA_GetPDNAddress(p_IPv4AllocationViaNasFlag, p_PdnIndex);
    
    f_EUTRA_MobileInfo_SetEPS_BearerContext(p_EpsDefaultBearerId, p_PduSessionInfo); // @sic R5-185166 sic@
    
    f_EUTRA_RRCConnectionReconfigurationWithAttachAcceptAllParams(p_CellId,
                                                                  p_EpsDefaultBearerId,    // @sic R5-144797 sic@
                                                                  tsc_RRC_TI_Def,
                                                                  p_EPS_TI,
                                                                  v_AttachType,
                                                                  v_PDN_Address,
                                                                  v_Guti,
                                                                  cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}),
                                                                  v_LAI,
                                                                  v_MSId,
                                                                  v_ESM_Cause, // @sic R5-104703 sic@
                                                                  p_APN,
                                                                  p_PcoToUE,
                                                                  f_GetAdditionalUpdateResult(p_AdditionalUpdateType, p_ForcedAttach), // @sic R5s110176 sic@
                                                                  p_T3412,
                                                                  p_T3402,
                                                                  p_T3423,
                                                                  p_EquivalentPlmnList,
                                                                  v_EmergNumList,         // @sic R5s130681 sic@
                                                                  -,
                                                                  -,
                                                                  p_BearerContextNumber, // @sic R5-144797 sic@
                                                                  p_EMM_Cause, // @sic R5-145971 sic@
                                                                  omit,
                                                                  p_DCN_Id, // @sic R5-177082  sic@
                                                                  p_ExtdPco); // @sic R5-200649 sic@
  }

	  function f_EUTRA_InitialRegistration_Step14_15(EUTRA_CellId_Type p_CellId,
                                                 EUTRA_ATTACH_Type p_ForcedAttach,
                                                 template (omit) AdditionalUpdateType p_AdditionalUpdateType,
                                                 ProcedureTransactionIdentifier p_EPS_TI,
                                                 boolean p_IPv4AllocationViaNasFlag,
                                                 template (omit) ProtocolConfigOptions p_PcoToUE,
                                                 AccessPointName p_APN,
                                                 template (value) GPRS_Timer p_T3412 := cs_GprsTimer_v_deact,
                                                 template (omit) GPRS_Timer p_T3402 := omit,
                                                 template (omit) GPRS_Timer p_T3423 := omit,
                                                 template (omit) PLMN_List p_EquivalentPlmnList := omit,
                                                 integer p_NoOfEmergencyNumbers := 0, // @sic R5-115673 sic@
                                                 template (omit) EMM_Cause p_EMM_Cause := omit,
                                                 EUTRA_InitialRegistrationPDN_Type p_InitialRegistrationPDN := firstPDN,
                                                 template (omit) DCN_ID p_DCN_Id := omit,
                                                 template (omit) EUTRA_NR_PduSessionInfo_Type p_PduSessionInfo := omit,
                                                 template (omit) ExtdProtocolConfigOptions p_ExtdPco := omit) runs on EUTRA_PTC
  { /* @sic R5-112600: IP address allocation sic@ */
    /* @sic R5-110796: split into f_EUTRA_InitialRegistration_Step5_13 and f_EUTRA_InitialRegistration_Step14_15 sic@ */
    /* @sic R5-120234, R5-120541, R5-120542 sic@ */
    /* @sic R5-144370, R5-144797: p_APN shall not be omit sic@ */
    /* @sic R5-145971: new parameter p_EMM_Cause sic@ */
    /* @sic R5s150510: new parameter p_InitialRegistrationPDN sic@ */
    /* @sic R5s170307: p_PDN_TypeToBeUsed removed sic@ */
    /* @sic R5s180648: got NetworkFeatureSupport from global variable sic@*/
    /* @sic R5-199103: p_PduSessionInfo sic@ */

    var PDN_TypeToBeUsed_Type v_PDN_TypeToBeUsed := f_EUTRA_MobileInfo_GetPdnTypeToBeUsed();   /* @sic R5s170307 sic@ */
    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(p_ForcedAttach); // @sic R5-103673, R5s100795 sic@
    
    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(p_CellId);
    var template (value) MobileIdentity v_Guti := f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams);
    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);
    
    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode(p_CellId);
    var template (omit) LocAreaId v_LAI := f_GetLAI(v_PLMN, v_LAC, p_ForcedAttach); // @sic R5s110176 sic@
    var TrackingAreaCode v_TAC := f_EUTRA_CellInfo_GetTAC(p_CellId);
    
    var template (omit) EmergNumList v_EmergNumList := omit;
    var template (omit) MobileIdentity v_MSId := f_GetMSId(p_ForcedAttach); // @sic R5-103673 sic@
    var PDN_Index_Type v_PdnIndex := f_EUTRA_InitialRegistrationPdnIndex(p_InitialRegistrationPDN);                   // @sic R5-144797, R5s150510, R5s180526 sic@
    var EPS_BearerIdentity v_EpsDefaultBearerId := f_EUTRA_InitialRegistrationEpsBearerId(p_InitialRegistrationPDN);  // @sic R5-144797, R5s150510, R5-183247 sic@
    var template (omit) ESM_Cause v_ESM_Cause := omit;        // @sic R5s120136 sic@
    var template (value) PDN_Address v_PDN_Address;
    var EmergencyNumList v_EmergencyNumList;
    
    var BEARER_CONTEXT_TYPE v_BearerContextNumber := DEF_1;
    
    if (pc_IPv6 and pc_IPv4) {              /* @sic R5-120234, R5-120541, R5-120542 sic@ */
      select (v_PDN_TypeToBeUsed) {
        case (forceIPv4only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110010'B); }     // PDN type IPv4 only allowed
        case (forceIPv6only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110011'B); }     // PDN type IPv6 only allowed
        case (pdnTypeAsSupportedByUE) { v_ESM_Cause := omit; }
      }
    }
    
    if (p_NoOfEmergencyNumbers > 0) {
      v_EmergencyNumList := f_Get_EmergencyNumList (p_NoOfEmergencyNumbers, f_EUTRA_MobileInfo_GetLocalEmergencyNumberList()); /* @sic R5s141315 sic@ */
      f_EUTRA_MobileInfo_SetLocalEmergencyNumberList (v_EmergencyNumList); // save the list of locally usable emergency numbers
      v_EmergNumList := f_Build_EmergNumList(p_NoOfEmergencyNumbers, v_EmergencyNumList);
    }
    if (tsc_EUTRA_Registration_PDN1isIMS) {     // @sic R5-144797, R5s141367 sic@
      v_BearerContextNumber := DEF_2;
    }
    
    v_PDN_Address := f_EUTRA_GetPDNAddress(p_IPv4AllocationViaNasFlag, v_PdnIndex);
    
    f_EUTRA_MobileInfo_SetEPS_BearerContext(v_EpsDefaultBearerId, p_PduSessionInfo); // @sic R5-185166 sic@
    
    f_EUTRA_RRCConnectionReconfigurationWithAttachAcceptAllParams(p_CellId,
                                                                  v_EpsDefaultBearerId,    // @sic R5-144797 sic@
                                                                  tsc_RRC_TI_Def,
                                                                  p_EPS_TI,
                                                                  v_AttachType,
                                                                  v_PDN_Address,
                                                                  v_Guti,
                                                                  cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}),
                                                                  v_LAI,
                                                                  v_MSId,
                                                                  v_ESM_Cause, // @sic R5-104703 sic@
                                                                  p_APN,
                                                                  p_PcoToUE,
                                                                  f_GetAdditionalUpdateResult(p_AdditionalUpdateType, p_ForcedAttach), // @sic R5s110176 sic@
                                                                  p_T3412,
                                                                  p_T3402,
                                                                  p_T3423,
                                                                  p_EquivalentPlmnList,
                                                                  v_EmergNumList,         // @sic R5s130681 sic@
                                                                  -,
                                                                  -,
                                                                  v_BearerContextNumber, // @sic R5-144797 sic@
                                                                  p_EMM_Cause, // @sic R5-145971 sic@
                                                                  omit,
                                                                  p_DCN_Id, // @sic R5-177082  sic@
                                                                  p_ExtdPco); // @sic R5-200649 sic@

    f_EUTRA_InitialRegistration_Step14_15_Common(p_CellId,
                                                 p_ForcedAttach,
                                                 p_AdditionalUpdateType,
                                                 p_EPS_TI,
                                                 p_IPv4AllocationViaNasFlag,
                                                 p_PcoToUE,
                                                 p_APN,
                                                 v_PdnIndex,
                                                 v_EpsDefaultBearerId,
                                                 v_BearerContextNumber,
                                                 p_T3412,
                                                 p_T3402,
                                                 p_T3423,
                                                 p_EquivalentPlmnList,
                                                 p_NoOfEmergencyNumbers,
                                                 p_EMM_Cause,
                                                 p_DCN_Id,
                                                 p_PduSessionInfo,
                                                 p_ExtdPco);
  }



Change 2 – f_EUTRA_RRCConnectionReconfiguration_Common
f_EUTRA_RRCConnectionReconfiguration_Common split into f_EUTRA_RRCConnectionReconfiguration_Common and f_EUTRA_RRCConnectionReconfiguration_Send.
	  function f_EUTRA_RRCConnectionReconfiguration_Send(EUTRA_CellId_Type p_CellId,
                                                     RRC_TransactionIdentifier p_RRC_TI,
                                                     UInt_Type p_NoOfAddDRBs_UM := 0,
                                                     template (omit) SRB_ToAddModList p_SRB_ToAddModList := omit,
                                                     template (value) DRB_ToAddModList p_DRB_ToAddModList,
                                                     template (omit) MAC_MainConfig_Type p_MAC_MainConfig := omit,
                                                     template (omit) PhysicalConfigDedicated p_PhysicalConfigDedicated := omit,
                                                     template (omit) NAS_MSG_RequestList_Type p_NAS_MsgList := omit,
                                                     template (omit) RLF_TimersAndConstants_r9 p_RLF_TimersAndConstants_r9 := omit,
                                                     template (omit) DownlinkAntennaGroupConfig_Type p_DownlinkAntennaGroupConfig := omit,
                                                     template (omit) RRCConnectionReconfiguration_v890_IEs p_RRCConnectionReconfiguration_v890_IEs := omit) runs on EUTRA_PTC
  {
    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info (p_CellId); //Get FDD or TDD mode from Cell-configuration
    var template (value) MAC_MainConfig_Type v_MAC_MainConfig;
    var template (omit)  PhysicalConfigDedicated v_PhysicalConfigDedicated;
    var template (value) RadioResourceConfigDedicated  v_RadioResourceConfigDedicated;
    var template (value) DRX_Config v_DRX_Config;
    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (p_CellId); //@sic R5-100786 sic@
    var template (value) TimingInfo_Type v_TimingInfo := cs_TimingInfo_Now;   /* @sic R5-104796 DRX support sic@ */
    var integer v_NoOfNASPDUs;
    var SubFrameTiming_Type v_SubFrameTiming;

    if (isvalue(p_MAC_MainConfig)) {
      v_MAC_MainConfig := valueof(p_MAC_MainConfig);
    }
    else {
      if (p_NoOfAddDRBs_UM > 0) {                /* @sic R5-104796 DRX support sic@ */
        v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxS; // short DRX cycle acc. to note in 36.508 table 4.6.3-16
      }
      else {
        v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxL; // long DRX cycle acc. to note in 36.508 table 4.6.3-16
      }
    }

    if (isvalue(p_PhysicalConfigDedicated)) {
      v_PhysicalConfigDedicated := valueof(p_PhysicalConfigDedicated);
    } else {
      if (isvalue(p_SRB_ToAddModList)) {   // @sic R5-130491, R5-121736, R5-121737 - support of MO, MT video call: per default PhysicalConfigDedicated are needed only if SRB2 is configured too sic@
        v_PhysicalConfigDedicated := cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,
                                                                                v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,
                                                                                v_AntennaInfo);
      } else {
        v_PhysicalConfigDedicated := omit;
      }
    }

    v_RadioResourceConfigDedicated := cds_RadioResourceConfigDedicated_SrbDrbConfig(p_SRB_ToAddModList,           // @sic R5-130491, R5-121736, R5-121737 - support of MO, MT video call sic@
                                                                                    p_DRB_ToAddModList,
                                                                                    v_MAC_MainConfig,
                                                                                    v_PhysicalConfigDedicated,
                                                                                    p_RLF_TimersAndConstants_r9); // @sic R5s120808 sic@
 
    if (p_NoOfAddDRBs_UM > 0) {    /* @sic R5-104796 DRX support sic@ */
      if (ispresent(v_MAC_MainConfig.explicitValue.drx_Config)) {                /* @sic R5120095 sic@ */
        v_DRX_Config := valueof(v_MAC_MainConfig.explicitValue.drx_Config);      /* valueof is ok as v_MAC_MainConfig.explicitValue.drx_Config is checked with 'ispresent' beforehand */
        f_EUTRA_SS_DrxCtrlConfig(p_CellId, cs_DrxCtrl_Config(v_DRX_Config));     /* configure DRX at the SS; @sic R5s100630 R5s120054 sic@ */
        
        if (isvalue(p_NAS_MsgList)) { 
          v_NoOfNASPDUs := lengthof(valueof(p_NAS_MsgList));
          v_SubFrameTiming := f_EUTRA_GetNextSendOccasion(p_CellId, 100 + ((v_NoOfNASPDUs - 1) * 20));  /* @sic R5-115770 change 5 sic@ */ 
        } else {                                                                                        /* @sic R5s211690: when there is no NAS signalling, no additional delay is needed sic@ */
          v_SubFrameTiming := f_EUTRA_GetNextSendOccasion(p_CellId);
        }
        v_TimingInfo := cs_TimingInfo_EUTRA_NB(v_SubFrameTiming);
      }
    }
    else {
      /* do nothing; for the time being we don't need any DRX configuration at the SS for DRX_L */
    }

    /* Step 8 */
    SRB.send(cas_SRB1_RrcNasPduList_REQ(p_CellId,
                                        v_TimingInfo,
                                        cs_RRCConnectionReconfiguration_Common(p_RRC_TI,
                                                                               omit,
                                                                               omit,
                                                                               v_RadioResourceConfigDedicated,
                                                                               omit,
                                                                               p_RRCConnectionReconfiguration_v890_IEs), // @sic R5-175050 sic@
                                        p_NAS_MsgList));

    if (ischosen(v_AntennaInfo.explicitValue)) {
      f_EUTRA_SS_CommonCellConfig(p_CellId, cas_PhysicalLayerConfigDL_CcchDcchDtchDL_Config_REQ(p_CellId,
                                                                                                cs_TimingInfo_Now,
                                                                                                cs_PhysicalLayerConfigDL_AntennaInfo(p_DownlinkAntennaGroupConfig),
                                                                                                cs_DciDLInfo_Auto(cs_DciInfo_CcchDcchDtchDL_Dci1A(f_EUTRA_CellInfo_GetDL_ChBandwidth(p_CellId))),
                                                                                                cs_AntennaInfoDedicated_r8andLater(v_AntennaInfo.explicitValue)))
    }
    
  }

	  function f_EUTRA_RRCConnectionReconfiguration_Common(EUTRA_CellId_Type p_CellId,
                                                       RRC_TransactionIdentifier p_RRC_TI,
                                                       UInt_Type p_NoOfAddDRBs_AM := 0,
                                                       UInt_Type p_NoOfAddDRBs_UM := 0,
                                                       template (omit) SRB_ToAddModList p_SRB_ToAddModList := omit,
                                                       template (omit) DRB_ToAddModList p_DRB_ToAddModList := omit,
                                                       template (omit) MAC_MainConfig_Type p_MAC_MainConfig := omit,
                                                       template (omit) PhysicalConfigDedicated p_PhysicalConfigDedicated := omit,
                                                       template (omit) EPS_QoSList_Type p_QoS := omit,
                                                       template (omit) Tft_List_Type p_Tft := omit,
                                                       boolean p_PDCP_StatusReportRequired := true,
                                                       template (omit) RLF_TimersAndConstants_r9 p_RLF_TimersAndConstants_r9 := omit,
                                                       template (omit) DownlinkAntennaGroupConfig_Type p_DownlinkAntennaGroupConfig := omit,
                                                       template (omit) DRB_Identity p_DrbFor2ndPDN := omit,
                                                       template (omit) RRCConnectionReconfiguration_v890_IEs p_RRCConnectionReconfiguration_v890_IEs := omit,
                                                       template (omit) UplinkDataCompression_r15_Type p_UDC_r15 := omit) runs on EUTRA_PTC
  { /* @sic R5-130491, R5-121736, R5-121737 - support of MO, MT video call: new parameter p_SRB_ToAddModList; function renamed to f_EUTRA_RRCConnectionReconfiguration_Common sic@ */
    /* @sic R5-144797: new parameter p_DrbFor2ndPDN sic@ */
    var integer v_DRB_Id;
    var integer i;
    var integer m;
    var integer n;
    var DRB_Identity v_DrbFor2ndPDN;
    var EPS_BearerIdentity v_EpsBearerId;
    var integer v_NoOfDefaultBearers := 1;  /* @sic R5-144797 sic@ */
    var integer v_NoOfQos := 0;
    var integer v_NoOfTft := 0;
    var integer v_NoOfAddDRBs := p_NoOfAddDRBs_AM + p_NoOfAddDRBs_UM;  /* number of bearers to be configured additionally (i.e. apart from the default bearer)
                                                                          if v_NoOfAddDRBs=0 => only the default DRB is established */
    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info (p_CellId); //Get FDD or TDD mode from Cell-configuration
    var template (value) DRB_ToAddModList v_DRB_ToAddModList;
    var template (value) MAC_MainConfig_Type v_MAC_MainConfig;
    var template (omit)  PhysicalConfigDedicated v_PhysicalConfigDedicated;
    var template (omit)  NAS_MSG_RequestList_Type v_NAS_MsgList;
    var template (value) RadioResourceConfigDedicated  v_RadioResourceConfigDedicated;
    var template (value) DRX_Config v_DRX_Config;
    var template (value) PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (p_CellId); //@sic R5-100786 sic@
    var template (value) TimingInfo_Type v_TimingInfo := cs_TimingInfo_Now;   /* @sic R5-104796 DRX support sic@ */
    var SubFrameTiming_Type v_SubFrameTiming;

    if (isvalue(p_DrbFor2ndPDN)) { v_NoOfDefaultBearers := v_NoOfDefaultBearers + 1; }  /* @sic R5-144797 sic@ */

    // -----------------------------------------------------
    // Check parameters are valid
    if (isvalue(p_QoS)) {
      v_NoOfQos := lengthof(p_QoS);
    }
    if (isvalue(p_Tft)) {
      v_NoOfTft := lengthof(p_Tft);
    }
    if (v_NoOfQos != v_NoOfAddDRBs) {
      FatalError (__FILE__, __LINE__, "invalid no. of QoS");
    }
    if (v_NoOfTft != v_NoOfAddDRBs) {
      FatalError (__FILE__, __LINE__, "invalid no. of TFT");
    }

    // -----------------------------------------------------
    // initialise v_DRB_ToAddModList

    if (isvalue(p_DRB_ToAddModList)) {                  // explicit setting for DRB_ToAddMod
      if (lengthof(p_DRB_ToAddModList) != v_NoOfAddDRBs + v_NoOfDefaultBearers) {
        FatalError (__FILE__, __LINE__, "invalid no. of DRB_ToAddMod");
      }
      v_DRB_ToAddModList := valueof(p_DRB_ToAddModList);
    } else {                                            // use default configuration
      v_DRB_Id := tsc_DRB1;                             // default bearer
      for (n := 0; n < (p_NoOfAddDRBs_AM + 1); n := n + 1) {                // default bearer included in the list (=> p_NoOfAddDRBs_AM + 1)
        if (p_PDCP_StatusReportRequired == true) {
          if (isvalue(p_UDC_r15)) {
            v_DRB_ToAddModList[n] := cs_DRB_ToAddMod_UDC (v_DRB_Id, p_UDC_r15);
          } else {
            v_DRB_ToAddModList[n] := cs_508_DRB_ToAddMod_DEFAULT_AM(v_DRB_Id);
          }
        }
        else {
          if (isvalue(p_UDC_r15)) {FatalError (__FILE__, __LINE__, "UDC not foreseen with No Status Report");}
          v_DRB_ToAddModList[n] := cs_DRB_ToAddMod_NoPDCP_StatusReport(v_DRB_Id);
        }
        v_DRB_Id := v_DRB_Id + 1;
      }
      for (m := 0; m < p_NoOfAddDRBs_UM; m := m + 1) {
        v_DRB_ToAddModList[n+m] := cs_508_DRB_ToAddMod_DEFAULT_UM (v_DRB_Id);
        v_DRB_Id := v_DRB_Id + 1;
      }
    }
    if (isvalue(p_DrbFor2ndPDN)) {  /* @sic R5-144797 sic@ */
      v_DrbFor2ndPDN := valueof(p_DrbFor2ndPDN);
      if (v_DrbFor2ndPDN < v_DRB_Id) {
        FatalError (__FILE__, __LINE__, "2nd default bearer is within the range of dedicated bearers");
      }
      v_DRB_ToAddModList[v_NoOfAddDRBs+1] := cs_508_DRB_ToAddMod_DEFAULT_AM(v_DrbFor2ndPDN);
    }

    // -----------------------------------------------------
    // initialise RRCConnectionReconfiguration
    
    if (isvalue(p_MAC_MainConfig)) {
      v_MAC_MainConfig := valueof(p_MAC_MainConfig);
    }
    else {
      if (p_NoOfAddDRBs_UM > 0) {                /* @sic R5-104796 DRX support sic@ */
        v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxS; // short DRX cycle acc. to note in 36.508 table 4.6.3-16
      }
      else {
        v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxL; // long DRX cycle acc. to note in 36.508 table 4.6.3-16
      }
    }

    if (isvalue(p_PhysicalConfigDedicated)) {
      v_PhysicalConfigDedicated := valueof(p_PhysicalConfigDedicated);
    } else {
      if (isvalue(p_SRB_ToAddModList)) {   // @sic R5-130491, R5-121736, R5-121737 - support of MO, MT video call: per default PhysicalConfigDedicated are needed only if SRB2 is configured too sic@
        v_PhysicalConfigDedicated := cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,
                                                                                v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,
                                                                                v_AntennaInfo);
      } else {
        v_PhysicalConfigDedicated := omit;
      }
    }

    v_RadioResourceConfigDedicated := cds_RadioResourceConfigDedicated_SrbDrbConfig(p_SRB_ToAddModList,           // @sic R5-130491, R5-121736, R5-121737 - support of MO, MT video call sic@
                                                                                    v_DRB_ToAddModList,
                                                                                    v_MAC_MainConfig,
                                                                                    v_PhysicalConfigDedicated,
                                                                                    p_RLF_TimersAndConstants_r9); // @sic R5s120808 sic@
    // -----------------------------------------------------
    // initialise NAS messages

    if (v_NoOfAddDRBs == 0) {
      v_NAS_MsgList := omit;                          // no NAS messages to be sent
    } else {
      for (i := 0; i < v_NoOfAddDRBs; i := i+1) {     // only for the additional DRBs (there is no ActDedEPSBearerCxtReq for the already existing default bearer)
        // get EpsBearerId from the corresponding DRB_ToAddModList; DRB_ToAddModList[0] is the default bearer which needs to be skipped
        v_EpsBearerId := int2hex(valueof(v_DRB_ToAddModList[i+1].eps_BearerIdentity), 1);  // i+1 to skip default DRB; valueof cannot be avoided here

        v_NAS_MsgList[i] := cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                           cs_ActDedEPSBearerCxtReq_QoS(v_EpsBearerId, valueof(p_QoS[i]), valueof(p_Tft[i])));    // valueof to avoid compiler warnings: neither p_QoS nor p_Tft can be omit as v_NoOfAddDRBs > 0
      }
    }

    f_EUTRA_RRCConnectionReconfiguration_Send(p_CellId,
                                              p_RRC_TI,
                                              p_NoOfAddDRBs_UM,
                                              p_SRB_ToAddModList,
                                              v_DRB_ToAddModList,
                                              p_MAC_MainConfig,
                                              p_PhysicalConfigDedicated,
                                              v_NAS_MsgList,
                                              p_RLF_TimersAndConstants_r9,
                                              p_DownlinkAntennaGroupConfig,
                                              p_RRCConnectionReconfiguration_v890_IEs);

    if (p_NoOfAddDRBs_UM > 0) {    /* @sic R5-104796 DRX support sic@ */
      if (ispresent(v_MAC_MainConfig.explicitValue.drx_Config)) {                /* @sic R5120095 sic@ */
        v_DRX_Config := valueof(v_MAC_MainConfig.explicitValue.drx_Config);      /* valueof is ok as v_MAC_MainConfig.explicitValue.drx_Config is checked with 'ispresent' beforehand */
        f_EUTRA_SS_DrxCtrlConfig(p_CellId, cs_DrxCtrl_Config(v_DRX_Config));     /* configure DRX at the SS; @sic R5s100630 R5s120054 sic@ */
        
        v_SubFrameTiming := f_EUTRA_GetNextSendOccasion (p_CellId, 100 + ((v_NoOfAddDRBs - 1) * 20));  /* @sic R5-115770 change 5 sic@ */
        v_TimingInfo := cs_TimingInfo_EUTRA_NB(v_SubFrameTiming);
      }
    }
    else {
      /* do nothing; for the time being we don't need any DRX configuration at the SS for DRX_L */
    }

    /* Step 8 */
    SRB.send(cas_SRB1_RrcNasPduList_REQ(p_CellId,
                                        v_TimingInfo,
                                        cs_RRCConnectionReconfiguration_Common(p_RRC_TI,
                                                                               omit,
                                                                               omit,
                                                                               v_RadioResourceConfigDedicated,
                                                                               omit,
                                                                               p_RRCConnectionReconfiguration_v890_IEs), // @sic R5-175050 sic@
                                        v_NAS_MsgList));

    if (ischosen(v_AntennaInfo.explicitValue)) {
      f_EUTRA_SS_CommonCellConfig(p_CellId, cas_PhysicalLayerConfigDL_CcchDcchDtchDL_Config_REQ(p_CellId,
                                                                                                cs_TimingInfo_Now,
                                                                                                cs_PhysicalLayerConfigDL_AntennaInfo(p_DownlinkAntennaGroupConfig),
                                                                                                cs_DciDLInfo_Auto(cs_DciInfo_CcchDcchDtchDL_Dci1A(f_EUTRA_CellInfo_GetDL_ChBandwidth(p_CellId))),
                                                                                                cs_AntennaInfoDedicated_r8andLater(v_AntennaInfo.explicitValue)))
    }
    
  }



As consequence of the above changes there are further changes to hand over a DRB list instead of a single DRB in 
· f_EUTRA_InitialRegistration_ReleaseInternetPDN
· f_EUTRA38_ENDC_UE_DetachMO
· f_EUTRA_IPCAN_XCAP_Signalling
· fl_EUTRA_RbEst_With2ndDefPDN
· f_EUTRA_OptionalMultiPDN_IMSRegistration
· f_EUTRA_508_ProseCommunicationProcedure
Change 3 – f_Get_508_EPSQoS
Configuration DEF_3 has not been implemented yet
	  function f_Get_508_EPSQoS(template (omit) BEARER_CONTEXT_TYPE p_BearerContextNumber) return template (value) EPS_QualityOfService
  { /* @sic R5-144797: parameter p_BearerContextNumber sic@ */
    var template (value) EPS_QualityOfService v_QoS;
    var BEARER_CONTEXT_TYPE v_BearerContextNumber := fl_BearerContextNumber_Def(p_BearerContextNumber);
    
    select (v_BearerContextNumber){   /* @sic R5-144797: v_QoS does not depend on pc_IMS anymore sic@ */
      case (DEF_1) {
        v_QoS := cs_508_EPS_QoS_Default_1;        /* Qos acc. to 36.508 Table 6.6.1-1 column 1 */
      }
      case (DEF_3) {
        v_QoS := cds_508_EPS_QoS_Default_3;       /* Qos acc. to 36.508 Table 6.6.1-1 column 3 */
      }
      case else {
        v_QoS := cds_EPS_QoS_Def_nonGBR_QCI;      /* Qos acc. to 36.508 Table 6.6.1-1 column 2 */
      }
    }
    return v_QoS;
  }

	  template (value) EPS_QualityOfService cds_508_EPS_QoS_Default_3
  modifies cs_508_EPS_QoS_Default_1 :=
  {
    qci     := '01000101'B // 69 for MCPTT
  };



Change 4 – f_EUTRA_ActivateDedicatedEpsBearer_Speech, f_EUTRA_ActivateDedicatedEpsBearer_SpeechVideo, f_EUTRA_508_ActDedicatedEPSBearerCxtReq_Speech
f_EUTRA_508_ActDedicatedEPSBearerCxtReq_Speech needs to distinguish Reference dedicated EPS bearer context #4 and #5 of 36.508 Table 6.6.2-1
	
  type enumerated BEARER_CONTEXT_TYPE {                /* @status    APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NBIOT, NR5GC, NR5GC_IRAT, POS) */
    DEF_1, // EPS Default Bearer#1 defined in the first column of table 6.6.1
    DEF_2, // EPS Default Bearer#2 defined in the second column of table 6.6.1
    DEF_3, // EPS Default Bearer#3 defined in the third column of table 6.6.1
    DED_1, // EPS Dedicated Bearer#1 defined in the first column of table 6.6.2
    DED_2, // EPS Dedicated Bearer#2 defined in the second column of table 6.6.2
    DED_3, // EPS Dedicated Bearer#3 defined in the third column of table 6.6.2
    DED_4,  // EPS Dedicated Bearer#4 defined in the fourth column of table 6.6.2 @sic R5-123632 sic@
    DED_5  // EPS Dedicated Bearer#5 defined in the fifth column of table 6.6.2
  };

	  function f_EUTRA_ActivateDedicatedEpsBearer_Speech(EUTRA_CellId_Type p_CellId,
                                                     EPS_BearerIdentity p_EpsDefaultBearerId := tsc_EpsDefaultBearerId,
                                                     EPS_BearerIdentity p_EpsDedicatedBearerId := tsc_EpsDedicatedBearerId,
                                                     PDN_Index_Type p_PdnIndex := PDN_1,
                                                     boolean p_IsSRVCC := false,
                                                     BEARER_CONTEXT_TYPE p_BearerContextNumber := DED_4) runs on EUTRA_PTC
  {
    var DRB_Identity v_DrbId := f_EUTRA_NB_EpsBearerAssociatedDRB(p_EpsDedicatedBearerId);
    var template (value) IP_DrbInfo_Type v_DrbInfo := cs_DrbInfo_EUTRA(p_CellId, v_DrbId); /* @sic R5s160131 sic@ */
    var template (value) MAC_MainConfig_Type v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxS;
    var template (value) DRB_ToAddModList v_DRB_ToAddModList := {cs_508_DRB_ToAddMod_DEFAULT_UM(v_DrbId)};
    var template (value) NAS_DL_Message_Type v_ActDedicatedEPSBearerCxtReq := f_EUTRA_508_ActDedicatedEPSBearerCxtReq_Speech(p_EpsDefaultBearerId, p_EpsDedicatedBearerId, p_PdnIndex, p_IsSRVCC, p_BearerContextNumber);
    var template (value) NAS_MSG_RequestList_Type v_NAS_MsgList := {cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_ActDedicatedEPSBearerCxtReq)};

    f_EUTRA_SS_RtpRtcpLoopback(p_PdnIndex, tsc_IMS_MediaPort_M1, v_DrbInfo);  /* @sic R5-153746: RTP/RTCP loopback; R5s160131: v_DrbInfo sic@ */

    fl_EUTRA_ActivateDedicatedEpsBearer_UM(p_CellId, v_DRB_ToAddModList, v_MAC_MainConfig, v_NAS_MsgList);
  }

	  function f_EUTRA_ActivateDedicatedEpsBearer_SpeechVideo(EUTRA_CellId_Type p_CellId,
                                                          EPS_BearerIdentity p_EpsDefaultBearerId := tsc_EpsDefaultBearerId,
                                                          EPS_BearerIdentity p_EpsDedicatedBearerIdSpeech := tsc_EpsDedicatedBearerId,
                                                          EPS_BearerIdentity p_EpsDedicatedBearerIdVideo := tsc_EpsDedicatedBearerId2,
                                                          PDN_Index_Type p_PdnIndex := PDN_1,
                                                          BEARER_CONTEXT_TYPE p_BearerContextNumber := DED_4) runs on EUTRA_PTC
  {
    var DRB_Identity v_DrbIdSpeech := f_EUTRA_NB_EpsBearerAssociatedDRB(p_EpsDedicatedBearerIdSpeech); /* @sic R5s160131 sic@ */
    var DRB_Identity v_DrbIdVideo := f_EUTRA_NB_EpsBearerAssociatedDRB(p_EpsDedicatedBearerIdVideo);   /* @sic R5s160131 sic@ */
    var template (value) IP_DrbInfo_Type v_DrbInfoSpeech := cs_DrbInfo_EUTRA(p_CellId, v_DrbIdSpeech);
    var template (value) IP_DrbInfo_Type v_DrbInfoVideo := cs_DrbInfo_EUTRA(p_CellId, v_DrbIdVideo);
    var template (value) MAC_MainConfig_Type v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxS;
    var template (value) DRB_ToAddModList v_DRB_ToAddModList := {
      cs_508_DRB_ToAddMod_DEFAULT_UM(v_DrbIdSpeech),
      cs_508_DRB_ToAddMod_DEFAULT_UM(v_DrbIdVideo)
    };
    var template (value) NAS_DL_Message_Type v_ActDedicatedEPSBearerCxtReqSpeech := f_EUTRA_508_ActDedicatedEPSBearerCxtReq_Speech(p_EpsDefaultBearerId, p_EpsDedicatedBearerIdSpeech, p_PdnIndex, -, p_BearerContextNumber);
    var template (value) NAS_DL_Message_Type v_ActDedicatedEPSBearerCxtReqVideo := f_EUTRA_508_ActDedicatedEPSBearerCxtReq_Video(p_EpsDefaultBearerId, p_EpsDedicatedBearerIdVideo, p_PdnIndex);
    var template (value) NAS_MSG_RequestList_Type v_NAS_MsgList := {
      cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_ActDedicatedEPSBearerCxtReqSpeech),
      cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_ActDedicatedEPSBearerCxtReqVideo)
    };

    f_EUTRA_SS_RtpRtcpLoopback(p_PdnIndex, tsc_IMS_MediaPort_M1, v_DrbInfoSpeech);  /* @sic R5-153746: RTP/RTCP loopback; R5s160131: v_DrbInfo sic@ */
    f_EUTRA_SS_RtpRtcpLoopback(p_PdnIndex, tsc_IMS_MediaPort_M2, v_DrbInfoVideo);   /* @sic R5-153746: RTP/RTCP loopback; R5s160131: v_DrbInfo sic@ */

    f_EUTRA_ActivateDedicatedEpsBearer_UM(p_CellId, v_DRB_ToAddModList, v_MAC_MainConfig, v_NAS_MsgList);
  }

	  function f_EUTRA_508_ActDedicatedEPSBearerCxtReq_Speech(EPS_BearerIdentity p_EpsDefaultBearerId := tsc_EpsDefaultBearerId,
                                                          EPS_BearerIdentity p_EpsDedicatedBearerId := tsc_EpsDedicatedBearerId,
                                                          PDN_Index_Type p_PdnIndex := PDN_1,
                                                          boolean p_IsSRVCC := false,
                                                          BEARER_CONTEXT_TYPE p_BearerContextNumber := DED_4)
    runs on EUTRA_PTC return template (value) NAS_DL_Message_Type
  {
    var template (value) LinkedEpsBearerIdentity v_LinkedId := cs_LinkedId(p_EpsDefaultBearerId);
    var template (value) EPS_QualityOfService v_QoS;
    var template (value) TrafficFlowTemplate v_Tft;
    var template (value) EPS_QualityOfService v_QoS := cds_EPS_QoS_Dedicated_4;      /* EPS bearer context #4 (QCI 1) according to 36.508 table 6.6.2-1; @sic R5s130897 change 5 sic@ */
    var template (value) TrafficFlowTemplate v_Tft := f_EUTRA_Get_508_TrafficFlowTemplate(4, p_EpsDedicatedBearerId, -, p_PdnIndex);  /* EPS bearer context #4 (QCI 1) according to 36.508 table 6.6.2-1; @sic R5s130897 change 5 sic@ */
    var template (value) NAS_DL_Message_Type v_ActDedicatedEPSBearerCxtReq;
    
    select (p_BearerContextNumber) {
      case (DED_5) {
        v_QoS := cds_EPS_QoS_Dedicated_5;                                                        /* EPS bearer context #5 (QCI 65) for MCPTT according to 36.508 table 6.6.2-1*/
        v_Tft := f_EUTRA_Get_508_TrafficFlowTemplate(5, p_EpsDedicatedBearerId, -, p_PdnIndex);  /* EPS bearer context #5 (QCI 65) for MCPTT according to 36.508 table 6.6.2-1 */

      }
      case else {
        v_QoS := cds_EPS_QoS_Dedicated_4;                                                        /* EPS bearer context #4 (QCI 1) according to 36.508 table 6.6.2-1; @sic R5s130897 change 5 sic@ */
        v_Tft := f_EUTRA_Get_508_TrafficFlowTemplate(4, p_EpsDedicatedBearerId, -, p_PdnIndex);  /* EPS bearer context #4 (QCI 1) according to 36.508 table 6.6.2-1; @sic R5s130897 change 5 sic@ */
      }
    }

    if (p_IsSRVCC) {
      v_ActDedicatedEPSBearerCxtReq := cs_508_ActDedicatedEPSBearerCxtReq_SRVCC(p_EpsDedicatedBearerId, tsc_PTI_Unassigned, v_LinkedId, v_QoS, v_Tft);
    } else {
      v_ActDedicatedEPSBearerCxtReq := cs_508_ActDedicatedEPSBearerCxtReq(p_EpsDedicatedBearerId, tsc_PTI_Unassigned, v_LinkedId, v_QoS, v_Tft);
    }
    return v_ActDedicatedEPSBearerCxtReq;
  }

	  function f_EUTRA_NB_Get_508_TrafficFlowTemplate(integer p_BearerContext,
                                                  EPS_BearerIdentity p_EPS_BearerId,
                                                  B3_Type p_PdnTypeValue,
                                                  PDN_Index_Type p_PdnIndex := PDN_1) return template (value) TrafficFlowTemplate
  { /* @sic R5-112600: to replace cs_508_Tft_Dedicated_1 and cs_508_Tft_Dedicated_2 sic@ */
    var integer v_PacketFilterLen1ByRef;
    var integer v_PacketFilterLen2ByRef;
    var template (value) PacketFilter v_PacketFilter1ByRef;
    var template (value) PacketFilter v_PacketFilter2ByRef;
    var template (value) TrafficFlowTemplate.packetFilterList v_PacketFilterList;
    var integer v_IeLength;

    select (p_BearerContext) {
      case (1) {        // @sic R5-123632 sic@
        fl_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 1, p_EPS_BearerId, p_PdnTypeValue, p_PdnIndex);  // get packet filter #1 acc. to 36.508 Table 6.6.2-2 @sic R5s130495 sic@
        fl_Get_508_PacketFilter(v_PacketFilter2ByRef, v_PacketFilterLen2ByRef, 2, p_EPS_BearerId, p_PdnTypeValue, p_PdnIndex);  // get packet filter #2 acc. to 36.508 Table 6.6.2-3 @sic R5s130495 sic@
        
        v_IeLength := v_PacketFilterLen1ByRef + v_PacketFilterLen2ByRef + 1;  // packet filter 1 + packet filter 2 + one octet for tftOperationCode, eBit and noOfPktFilter
        v_PacketFilterList := {v_PacketFilter1ByRef, v_PacketFilter2ByRef};
      }
      case (2) {        // @sic R5-123632, R5-132085 sic@
        fl_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 3, p_EPS_BearerId, p_PdnTypeValue, p_PdnIndex);  // get packet filter #3 acc. to 36.508 Table 6.6.2-4 @sic R5s130495 sic@

        v_IeLength := v_PacketFilterLen1ByRef + 1;                           // packet filter 1 + one octet for tftOperationCode, eBit and noOfPktFilter
        v_PacketFilterList := {v_PacketFilter1ByRef};
      }
      case (3) {        // @sic R5-132085 sic@
        fl_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 5, p_EPS_BearerId, p_PdnTypeValue, p_PdnIndex);  // get packet filter #3 acc. to 36.508 Table 6.6.2-4 @sic R5s130495 sic@

        v_IeLength := v_PacketFilterLen1ByRef + 1;                           // packet filter 1 + one octet for tftOperationCode, eBit and noOfPktFilter
        v_PacketFilterList := {v_PacketFilter1ByRef};
      }
      case (4) {        // @sic R5-132085 sic@
        fl_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 4, p_EPS_BearerId, p_PdnTypeValue, p_PdnIndex);  // get packet filter #4 acc. to 36.508 Table 6.6.2-4 @sic R5s130495 sic@

        v_IeLength := v_PacketFilterLen1ByRef + 1;                           // packet filter 1 + one octet for tftOperationCode, eBit and noOfPktFilter
        v_PacketFilterList := {v_PacketFilter1ByRef};
      }
      case (5) {        // @sic R5-210201 sic@
        fl_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 4, p_EPS_BearerId, p_PdnTypeValue, p_PdnIndex);  // get packet filter #4 acc. to 36.508 Table 6.6.2-1
        fl_Get_508_PacketFilter(v_PacketFilter2ByRef, v_PacketFilterLen2ByRef, 10, p_EPS_BearerId, p_PdnTypeValue, p_PdnIndex); // get packet filter #10 acc. to 36.508 Table 6.6.2-1

        v_IeLength := v_PacketFilterLen1ByRef + v_PacketFilterLen2ByRef + 1;  // packet filter 1 + packet filter 2 + one octet for tftOperationCode, eBit and noOfPktFilter @sic R5s180640 sic@
        v_PacketFilterList := {v_PacketFilter1ByRef, v_PacketFilter2ByRef};
      }
      case (6) { // @sic R5-183239 sic@
        fl_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 6, p_EPS_BearerId, p_PdnTypeValue, p_PdnIndex);  // get packet filter #6 acc. to 36.508 Table 6.6.2-1A
        fl_Get_508_PacketFilter(v_PacketFilter2ByRef, v_PacketFilterLen2ByRef, 7, p_EPS_BearerId, p_PdnTypeValue, p_PdnIndex);  // get packet filter #7 acc. to 36.508 Table 6.6.2-1A

        v_IeLength := v_PacketFilterLen1ByRef + v_PacketFilterLen2ByRef + 1;  // packet filter 1 + packet filter 2 + one octet for tftOperationCode, eBit and noOfPktFilter @sic R5s180640 sic@
        v_PacketFilterList := {v_PacketFilter1ByRef, v_PacketFilter2ByRef}; // @sic R5s180640 sic@
      }
      case (7) {        // @sic R5-187765 sic@
        fl_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 8, p_EPS_BearerId, p_PdnTypeValue, p_PdnIndex);  // get packet filter #8 acc. to 36.508 Table 6.6.2-1A

        v_IeLength := v_PacketFilterLen1ByRef + 1;  // packet filter 1 + one octet for tftOperationCode, eBit and noOfPktFilter
        v_PacketFilterList := {v_PacketFilter1ByRef};
      }
      case (8) {        // @sic R5-187765 sic@
        fl_Get_508_PacketFilter(v_PacketFilter1ByRef, v_PacketFilterLen1ByRef, 9, p_EPS_BearerId, p_PdnTypeValue, p_PdnIndex);  // get packet filter #9 acc. to 36.508 Table 6.6.2-1A

        v_IeLength := v_PacketFilterLen1ByRef + 1;    // packet filter 1 + one octet for tftOperationCode, eBit and noOfPktFilter
        v_PacketFilterList := {v_PacketFilter1ByRef};
      }
      
      case else {
        FatalError(__FILE__, __LINE__, "invalid bearer context");
      }
    }
    return cs_TrafficFlowTemplate(v_IeLength, v_PacketFilterList);
  }

	  function fl_Get_508_PacketFilter(out template (value) PacketFilter p_PacketFilterByRef,
                                   out integer p_PacketFilterLengthByRef,
                                   integer p_PacketFilterNumber,
                                   HalfOctet_Type p_EPS_BearerId,
                                   B3_Type p_PdnTypeValue,
                                   PDN_Index_Type p_PdnIndex)
  { /* @sic R5-112600 new parameter p_PdnTypeValue to determine whether IPv4 or IPv6 address to be used sic@ */
    /* @sic R5s130362 - MCC160 Implementation: tsc_PDN_AddressInfo; new parameter p_PdnIndex sic@ */
    var integer v_ContentsLength := 0;
    var template (value) PacketFilter v_Filter;
    var PDN_AddressInfo_Type v_PDN_AddressInfo := f_PDN_AddressInfo_Get(p_PdnIndex);
    var charstring v_NetworkIpAddrv4 := v_PDN_AddressInfo.PCSCF_IPAddressIPv4;
    var charstring v_NetworkIpAddrv6 := v_PDN_AddressInfo.PCSCF_IPAddressIPv6;


    if ((p_PacketFilterNumber != 4) and (p_PacketFilterNumber != 5) and (p_PacketFilterNumber != 10)) {   // @sic R5-132085 sic@ in case (4, 5, 10) no addresses used @sic R5s130495 sic@
      if (p_PdnTypeValue == tsc_PdnType_IPv4) {  // IPv4 only; @sic R5s110168 change 8.3; R5-112600 sic@
        v_Filter.contents[0] := cs_PktFilterCompIPv4RemoteAddress(v_NetworkIpAddrv4);
        v_ContentsLength := v_ContentsLength + 9;
      }
      else { // IPv4v6 or IPv6 only
        v_Filter.contents[0] := cs_PktFilterCompIPv6RemoteAddress(v_NetworkIpAddrv6);
        v_ContentsLength := v_ContentsLength + 33;
      }
    }
    
    select (p_PacketFilterNumber){
      case (1, 6) { // @sic R5-183239 sic@
        v_Filter.spare       := '00'B; //  @sic R5s130195 Baseline Moving sic@
        v_Filter.direction   := '01'B; //  @sic R5s130195 Baseline Moving sic@
        v_Filter.iei         := '0'H;  //  @sic R5s130195 Baseline Moving sic@
        if (hex2int(p_EPS_BearerId) > 5) {
          v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8); // @sic R5-174469 sic@
        } else {
          v_Filter.precedence  := int2bit(0, 8);
        }
        if (p_PacketFilterNumber == 6) {  // @sic R5-183239, R5-193982 sic@
          if (p_PdnTypeValue == tsc_PdnType_IPv4) {  // IPv4 only
            v_Filter.contents[1] := cs_PktFilterCompProtocolIdNextHeaderICMP;
          } else {
            v_Filter.contents[1] := cs_PktFilterCompProtocolIdNextHeaderICMPv6;
          }
        } else {
          v_Filter.contents[1] := cs_PktFilterCompSingleRemotePort31160(p_EPS_BearerId);
          v_ContentsLength := v_ContentsLength + 3;
          v_Filter.contents[2] := cs_PktFilterCompProtocolIdNextHeaderUDP;
        }
        v_ContentsLength := v_ContentsLength + 2;
      }
      case (2, 7) { // @sic R5-183239 sic@
        // Update to values in Reference packet filter #2
        // UL only filter, ID=1          @sic R5-104702 sic@
        v_Filter.spare       := '00'B; //  @sic R5s130195 Baseline Moving sic@
        v_Filter.direction   := '10'B; //  @sic R5s130195 Baseline Moving sic@
        v_Filter.iei         := '1'H;  //  @sic R5s130195 Baseline Moving sic@
        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6+8), 8); // @sic R5-104702 sic@

        if (p_PacketFilterNumber == 7) {// @sic R5-183239, R5-193982 sic@
          if (p_PdnTypeValue == tsc_PdnType_IPv4) {  // IPv4 only
            v_Filter.contents[1] := cs_PktFilterCompProtocolIdNextHeaderICMP;
          } else {
            v_Filter.contents[1] := cs_PktFilterCompProtocolIdNextHeaderICMPv6;
          }
        } else {
          v_Filter.contents[1] := cs_PktFilterCompSingleRemotePort61000(p_EPS_BearerId);
          v_ContentsLength := v_ContentsLength + 3;
          v_Filter.contents[2] := cs_PktFilterCompProtocolIdNextHeaderUDP;
        }
        v_ContentsLength := v_ContentsLength + 2;
      }
      case (3){
        // Update to values in Reference packet filter #3
        // Bidirectional filter, ID=2  @sic R5-104702 sic@
        v_Filter.spare       := '00'B; //  @sic R5s130195 Baseline Moving sic@
        v_Filter.direction   := '11'B; //  @sic R5s130195 Baseline Moving sic@
        v_Filter.iei         := '2'H;  //  @sic R5s130195 Baseline Moving sic@
        v_Filter.precedence  := '00001111'B;
      }
      case (4){  // @sic R5-132085 sic@
        // Update to values in Reference packet filter #4
        // Bidirectional filter, ID=3
        v_Filter.spare       := '00'B;
        v_Filter.direction   := '11'B;
        v_Filter.iei         := '3'H;
        if (hex2int(p_EPS_BearerId)>5)
        {
          v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8); // @sic R5-174469 sic@
        } else {
          v_Filter.precedence  := int2bit(0, 8);
        }
        v_Filter.contents[0] := cs_PktFilterCompRemotePortRange(tsc_IMS_MediaPort_M1, tsc_IMS_MediaPort_M1+1);  // @sic R5s130495 sic@
        v_ContentsLength := v_ContentsLength + 5;
        v_Filter.contents[1] := cs_PktFilterCompProtocolIdNextHeaderUDP;
        v_ContentsLength := v_ContentsLength + 2;
      }
      case (5){  // @sic R5-132085 sic@
        // Update to values in Reference packet filter #5
        // Bidirectional filter, ID=3
        v_Filter.spare       := '00'B;
        v_Filter.direction   := '11'B;
        v_Filter.iei         := '4'H;
        if (hex2int(p_EPS_BearerId)>5)
        {
          v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8); // @sic R5-174469 sic@
        } else {
          v_Filter.precedence  := int2bit(0, 8);
        }
        v_Filter.contents[0] := cs_PktFilterCompRemotePortRange(tsc_IMS_MediaPort_M2, tsc_IMS_MediaPort_M2+1);  // @sic R5s130495 sic@
        v_ContentsLength := v_ContentsLength + 5;
        v_Filter.contents[1] := cs_PktFilterCompProtocolIdNextHeaderUDP;
        v_ContentsLength := v_ContentsLength + 2;
      }
      case (8) { // @sic R5-187765 sic@
        // Update to values in Reference packet filter #8
        // UL only filter, ID=1
        v_Filter.spare       := '00'B;
        v_Filter.direction   := '10'B;
        v_Filter.iei         := '1'H;
        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6+8), 8);
      }
      case (9) { // @sic R5-187765 sic@
        // Update to values in Reference packet filter #9
        // Bidirectional filter, ID=5
        v_Filter.spare       := '00'B;
        v_Filter.direction   := '11'B;
        v_Filter.iei         := '5'H;
        if (hex2int(p_EPS_BearerId)>5)
        {
          v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8);
        } else {
          v_Filter.precedence  := int2bit(0, 8);
        }
        v_Filter.contents[0] := cs_PktFilterCompRemotePortRange(tsc_IMS_MediaPort_M2, tsc_IMS_MediaPort_M2+1);
        v_ContentsLength := v_ContentsLength + 5;
        v_Filter.contents[1] := cs_PktFilterCompProtocolIdNextHeaderUDP;
        v_ContentsLength := v_ContentsLength + 2;
      }
      case (10) { // @sic R5-210201 sic@
        // Update to values in Reference packet filter #10
        // Bidirectional filter, ID=4
        v_Filter.spare       := '00'B;
        v_Filter.direction   := '11'B;          // Bidirectional
        v_Filter.iei         := '4'H;           // ID=4
        if (hex2int(p_EPS_BearerId)>5)
        {
          v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8);
        } else {
          v_Filter.precedence  := int2bit(0, 8);
        }
        v_Filter.contents[0] := cs_PktFilterCompSingleRemotePort(tsc_IMS_MediaPort_M3);  // same portnumber as tsc_MCX_MediaControlPortNW
        v_ContentsLength := v_ContentsLength + 3;
        v_Filter.contents[1] := cs_PktFilterCompProtocolIdNextHeaderUDP;
        v_ContentsLength := v_ContentsLength + 2;
      }
    }
    v_Filter.iel := int2oct(v_ContentsLength, 1);
    
    p_PacketFilterByRef := v_Filter;
    p_PacketFilterLengthByRef := v_ContentsLength + 3; // length of the filter contents + 3 octets for iei, precedence and iel
  }

